Program Verification: Lecture 13

José Meseguer

Computer Science Department
University of Illinois at Urbana-Champaign

The ITP Inference Rules

Notice that in the ITP we reason backwards, replacing the main goal G we want to prove by subgoals, $G_1, \ldots, G_n$, such that if we prove each of the subgoals, then we have proved the main goal.

For such an inference to be sound, the implication

$$G_1 \wedge \ldots \wedge G_n \Rightarrow G$$

should always be satisfied, that is, should be semantically valid in the initial algebra $\mathcal{T}_{\Sigma,E}$ on which we are doing the inductive reasoning.

The ITP Inference Rules (II)

Such semantically valid inferences are expressed as inference rules

$$\frac{G_1 \quad \ldots \quad G_n}{G}$$

However, since we are reasoning backwards, from the root of the proof tree to the leaves, the ITP uses such rules in the opposite direction, as rules

$$\frac{G}{G_1 \quad \ldots \quad G_n}$$

We will illustrate through an example such backward reasoning for several ITP inference rules besides the induction rule, and will at the same time justify their soundness.

Linearity of the Number Ordering

Consider the following module defining the order on numbers, which we would like to prove is linear.

```
fmod NATURAL-ORD is
  sort Natural .
  op 0 : -> Natural [ctor] .
  op s : Natural -> Natural [ctor] .
  op _<_ : Natural Natural -> Bool .
  op _=<_ : Natural Natural -> Bool .
  vars N M : Natural .
  eq N < 0 = false .
  eq 0 < s(N) = true .
  eq s(N) < s(M) = N < M .
  ceq N =< M = true if N < M .
  ceq N =< M = true if not(M < N) .
  ceq N =< M = false if M < N .
endfm
```

The cns Inference Rule

After entering the goal stating that the order on the naturals is linear, one of the possible ITP inference rules we can invoke is the lemma of constants, which converts universally quantified variables in a goal into constants.

```
Maude> (goal linear : NATURAL-ORD |- A{N:Natural ; M:Natural}
       (((N =< M) or (M =< N)) = (true)) .)

==================================
label-sel: linear@0
==================================
A{N:Natural ; M:Natural}
N:Natural =< M:Natural or M:Natural =< N:Natural = true
++++++++++++++++++++++++++++++++++

Maude> (cns .)
```

```
==================================
label-sel: linear@0
==================================
N*Natural =< M*Natural or M*Natural =< N*Natural = true
++++++++++++++++++++++++++++++++++
```

The `cns` Inference Rule (II)

The fact that this is a semantically valid inference is based on the **Constants Lemma**, which states the equivalence between satisfiability of a quantified equation, and of the same equation with the variables transformed into generic constants,

$$E \models_{\Sigma} t = t' \quad \Leftrightarrow \quad E \models_{\Sigma(X)} t = t'.$$

Thanks to the completeness of equational reasoning, this is expressed in the ITP as the `cns` rule,

$$\frac{E \vdash_{\Sigma} t = t'}{E \vdash_{\Sigma(X)} t = t'}$$

Justification of the cns Rule

We can justify the validity of the **Constants Lemma** not only for unconditional equations, but also for conditional ones as follows. Given a conditional equation

$$(\forall X)\ \varphi = (\forall X)\ t = t' \quad \Leftarrow \quad u_1 = v_1 \wedge \ldots \wedge u_n = v_n,$$

if $X = vars(\varphi)$, we have $E \models_\Sigma \varphi$ iff, by definition,

$$(\forall (\mathcal{A}, a) \in \mathbf{Alg}_{\Sigma(X)})\ \mathcal{A} \models_\Sigma E \quad \Rightarrow \quad (\mathcal{A}, a) \models_{\Sigma(X)} \varphi,$$

which is equivalent, by **Ex**10.1, to

$$(\forall (\mathcal{A}, a) \in \mathbf{Alg}_{\Sigma(X)})\ (\mathcal{A}, a) \models_{\Sigma(X)} E \quad \Rightarrow \quad (\mathcal{A}, a) \models_{\Sigma(X)} \varphi,$$

which, by definition of satisfaction, is precisely $E \models_{\Sigma(X)} \varphi$, as desired.

Reasoning by Cases: The `split` Rule

To prove our goal, we can now reason by cases. For each pair of natural numbers n, m, either $n < m = \texttt{true}$, or $n < m = \texttt{false}$. Therefore, we can split a goal involving n and m into two subgoals: one in which we assume $n < m = \texttt{true}$ as an extra hypothesis, and another in which we assume $n < m = \texttt{false}$.

In the ITP this is accomplished by the `split` rule, which, given an unquantified goal without variables (only "generic" constants like n, m) and given a Boolean-valued expression involving some of those generic constants, splits a given goal into two: one assuming the expression `true`, and another assuming it `false`.

Reasoning by Cases: The `split` Rule (II)

In our example we can give the `split` command,

```
Maude> (split on (N*Natural < M*Natural) . )

==================================
label-sel: linear@1.0
==================================
N*Natural =< M*Natural or M*Natural =< N*Natural = true

==================================
label: linear@2.0
==================================
N*Natural =< M*Natural or M*Natural =< N*Natural = true
++++++++++++++++++++++++++++++++++
```

Reasoning by Cases: The `split` Rule (III)

The goals remain the same, but the split hypotheses for each case have been added to each case's module. Using these hypotheses we can now discharge each of the subgoals with the `auto` tactic.

```
Maude> (auto .)

==================================
label-sel: linear@2.0
==================================
N*Natural =< M*Natural or M*Natural =< N*Natural = true
++++++++++++++++++++++++++++++++++

Maude> (auto .)

q.e.d

++++++++++++++++++++++++++++++++++
```

Caveats on the `split` Rule

As already mentioned, the present version of the ITP requires that the goal to which the `split` rule is applied is an unquantified goal without variables, in which only "generic" constants —such as `N*Natural` and `M*Natural` in our example— appear.

This requirement will be relaxed in future ITP versions, but it is assumed and required by the present version. Therefore, applications of `split` to quantified goals with variables are currently forbidden: in the present ITP version we must first transform such variables into generic constants using `cns`.

Applications of `split` Should Protect `BOOL`

Regardless of the current ITP restrictions in the application of `split`, there is a fundamental way in which the application of `split` would be unsound, namely, if we have messed up the Booleans by adding junk and perhaps confusion to them, so that we do not have anymore two different canonical forms, `true`, and `false`, but may have other nonstandard Boolean elements as well.

This can happen because, when defining new predicates with operators of sort `Bool`, we do not give enough equations, thus adding "junk" to `Bool`, or we give the wrong equations, adding "confusion" and possibly also "junk" to `Bool`. Therefore, for an application of `split` to be sound, it is enough to require that the `BOOL` submodule is protected.

Protecting Module Importations

In general, if we have a theory (Σ, E) having a subtheory (Σ', E') with,

$$\Sigma' \subseteq \Sigma \quad \text{and} \quad E' \subseteq E,$$

and the module `fmod`(Σ, E)`endfm` imports the submodule `fmod`(Σ', E')`endfm` in `protecting` mode, we require that the unique Σ'-homomorphism

$$eval^{E'}_{\mathcal{T}_{\Sigma/E}|_{\Sigma'}} : \mathcal{T}_{\Sigma'/E'} \longrightarrow \mathcal{T}_{\Sigma/E}|_{\Sigma'}$$

is an isomorphism.

Protecting Module Importations (II)

Checking that `BOOL` is protected in a supermodule `fmod`(Σ, E)`endfm` is quite easy, since it reduces to checking that:

1. In (Σ, E) the only constructors of sort `Bool` are `true` and `false`.

2. (Σ, E) is ground confluent, terminating, sort decreasing, and sufficiently complete relative to the given signature Ω of declared constructors; and

3. `true` and `false` are in E-canonical form.

Justification of the `split` Rule

Suppose that we are reasoning inductively about a module $\texttt{fmod}(\Sigma, E)\texttt{endfm}$, which correctly imports `BOOL` in `protecting` mode (**Note**: this must be checked independently, as an implicit proof obligation).

In `BOOL` we have,

$$\mathcal{T}_{\texttt{BOOL}} \models \texttt{true} \neq \texttt{false}$$

and we also have,

$$\mathcal{T}_{\texttt{BOOL}} \models (\forall x : \texttt{Bool})\ x = \texttt{true} \vee x = \texttt{false}$$

Notice that, by the `protecting` importation, we have, $\mathcal{T}_{\Sigma/E}|_{\Sigma_{\texttt{BOOL}}} \cong \mathcal{T}_{\texttt{BOOL}}$.

Justification of the `split` Rule (II)

Therefore, given any Boolean valued Σ-term $p \in T_\Sigma(X)_{\texttt{Bool}}$, and given any assignment $a : X \longrightarrow T_{\Sigma/E}$, we have,

$$(\mathcal{T}_{\Sigma/E}, a) \not\models (p \neq \texttt{true} \wedge p \neq \texttt{false}).$$

That is, for any such assignment a, the above proposition is equivalent to the identically false proposition, $\bot$, which is never satisfied:

$$(\mathcal{T}_{\Sigma/E}, a) \models (p \neq \texttt{true} \wedge p \neq \texttt{false}) \quad \Leftrightarrow \quad (\mathcal{T}_{\Sigma/E}, a) \models \bot.$$

Now, since for any proposition A we always have the Boolean equivalence, $A \equiv A \vee \bot$, we also have an equivalence $(\forall X)\ A \equiv (\forall X)\ (A \vee \bot)$. Therefore, given an equation $t = t'$ with $vars(t = t) \subseteq X$ we have,

Justification of the `split` Rule (III)

$$\mathcal{T}_{\Sigma/E} \models t = t' \quad \Leftrightarrow \quad \mathcal{T}_{\Sigma/E} \models ((t = t') \vee \bot)$$

or, equivalently,

$$\mathcal{T}_{\Sigma/E} \models (\forall X)\, t = t' \quad \Leftrightarrow \quad \mathcal{T}_{\Sigma/E} \models (\forall X)\, ((t = t') \vee (p \neq \texttt{true} \wedge p \neq \texttt{false}))$$

which, using distributivity of disjunction over conjunction, plus the fact that $A \Rightarrow B \equiv (\neg A) \vee B$, is equivalent to,

$$\mathcal{T}_{\Sigma/E} \models (\forall X)\, t = t' \quad \Leftrightarrow \quad \mathcal{T}_{\Sigma/E} \models (\forall X)\, (p = \texttt{true} \Rightarrow t = t') \wedge (p = \texttt{false} \Rightarrow t = t'),$$

which, modulo the distribution of $\forall$ over $\wedge$, is our desired justification for `split` as a sound inductive reasoning rule.

Induction on Other Data Structures: Tree Induction

We have already seen examples of how the ITP's `ind` rule applies to natural number induction and to list induction.

Before discussing the most general form of the `ind` rule for any signature of constructors Ω and its justification, we give an example illustrating binary tree induction, in which the data in leaves are seen as depth-zero trees.

The intuitive idea is that to prove an inductive property P about such trees we must show: (1) that P holds for the data elements (**base case**); and (2) that if P holds for the left and right subtrees, then it must hold for their binary join (**induction step**).

Induction on Other Data Structures: Tree Induction (II)

Consider the following module defining binary trees whose nodes are quoted identifiers (constants in the predefined module QID), and a reverse function on binary trees.

```
fmod TREE is
protecting QID .
sort Tree .
subsort Qid < Tree .
op _#_ : Tree Tree -> Tree [ctor] .
op rev : Tree -> Tree .
var I : Qid .
vars T T' : Tree .
eq rev(I) = I .
eq rev(T # T') = rev(T') # rev(T) .
endfm
```

Induction on Other Data Structures: Tree Induction (III)

We can apply binary tree induction to prove that for all trees `T` the equation `rev(rev(T)) = T` holds. We can do so by entering the `TREE` module in the ITP and the goal:

```
Maude> (goal rev : TREE |- A{T:Tree}((rev(rev(T:Tree))) = (T:Tree)) .)

==================================
label-sel: rev@0
==================================
A{T:Tree} rev(rev(T:Tree)) = T:Tree
++++++++++++++++++++++++++++++++++
```

Induction on Other Data Structures: Tree Induction (IV)

We can then try to prove this goal by induction on T:Tree.

```
Maude> (ind on T:Tree .)


==================================
label-sel: rev@1.0
==================================
A{V0#0:Qid} rev(rev(V0#0:Qid)) = V0#0:Qid ==>
rev(rev(V0#0:Qid)) = V0#0:Qid

==================================
label: rev@2.0
==================================
A{V0#0:Tree ; V0#1:Tree} rev(rev(V0#1:Tree)) = V0#1:Tree &
rev(rev(V0#0:Tree)) = V0#0:Tree ==>
rev(rev(V0#0:Tree # V0#1:Tree)) = V0#0:Tree # V0#1:Tree
++++++++++++++++++++++++++++++++++
```

Induction on Other Data Structures: Tree Induction (V)

Note that goal `rev@2.0` is the "induction step" in tree induction, whereas the "base case" is goal `rev@1.0`. Both subgoals can then be proved using the `auto` tactic.

```
Maude> (auto .)

==================================
label-sel: rev@2.0
==================================
A{V0#0:Tree ; V0#1:Tree} rev(rev(V0#1:Tree)) = V0#1:Tree &
rev(rev(V0#0:Tree)) = V0#0:Tree ==>
rev(rev(V0#0:Tree # V0#1:Tree)) = V0#0:Tree # V0#1:Tree
++++++++++++++++++++++++++++++++++

Maude> (auto .)

q.e.d
```

Structural Induction

We have already observed how the ITP supports inductive proofs in three cases: natural number induction, list induction, and tree induction. But what is the general form of induction supported by the ITP for a specification having a subsignature Ω of constructors? This general form is called structural induction. It reduces proving an inductive property of the form $(\forall x : s)\ P(x)$, to proving:

- **Base Case**. For any constant $a : nil \longrightarrow s'$ in Ω with $s' \leq s$, the subgoal $P(x/a)$

  Notation: Given a variable x, the substitution $\{(x, t)\}$ mapping x to a term t is denoted (x/t), and its homomorphic extension is denoted $\_(x/t)$.

Structural Induction (II)

- **Induction Step**. For each constructor $f : s_1 \ldots s_n \longrightarrow s'$ in Ω with $s' \leq s$, where the sorts $s_{i_1}, \ldots, s_{i_k}$ are those among the $s_1 \ldots s_n$ such that $s_{i_j} \leq s$, $1 \leq j \leq k$, the subgoal,

  $$(\forall x_1 : s_1, \ldots, x_n : s_n)\ P(x/x_{i_1}) \wedge \ldots \wedge P(x/x_{i_k}) \Rightarrow P(x/f(x_1, \ldots, x_n)).$$

  Note: It may happen that none of the sorts among the $s_1 \ldots s_n$ is s or a subsort of s. In that case, the subgoal takes the form $(\forall x_1 : s_1, \ldots, x_n : s_n)\ P(x/f(x_1, \ldots, x_n))$.

- **Subsorts Without Constructors**. If $s' \leq s$ is a subsort having no constructor constants or operators in a sort $s'' \leq s'$, then we add the subgoal $(\forall y : s')\ P(x/y)$.

Structural Induction (III)

Note: If the signature Ω of constructors has been fully specified, the base case and the induction step implicitly cover all subsorts, so that the third case should never arise, except perhaps for s' an empty sort, with no terms whatsoever, for which the property P then trivially holds.

Therefore, from now on we will systematically ignore the case of subsorts without constructors in the rest of our theoretical discussions. In practice, however, this case is actually quite useful, and this for two reasons:

Structural Induction (IV)

- to deal with constants in predeclared modules, such as `QID`, which are built-in and are not defined as constructors (we encountered this phenomenon in our tree-reverse example); and

- to generalize these inductive proof methods to parameterized modules, such as `LIST(X)`, where the parametric sort of elements might be a subsort of the sort `List(X)`, but we have no a priori information about the constructors of such a parametric sort of elements.

Structural Induction (V)

Ignoring the case of subsorts without constructors, this then becomes an inductive inference rule of the form,

$$\frac{\bigwedge_i P(x/a_i) \wedge \bigwedge_j (\forall \overline{x} : \overline{s})\ P(x/x_{i_1}) \wedge \ldots \wedge P(x/x_{i_k}) \Rightarrow P(x/f_j(x_1, \ldots, x_{n_j}))}{(\forall x : s)\ P(x)}$$

where the a_i and the f_j include all the constructor constants and operators meeting the properties specified above, and where $(\forall \overline{x} : \overline{s})$ abbreviates $(\forall x_1 : s_1, \ldots, x_{n_j} : s_{n_j})$.

Of course, in the ITP this rule is used backwards as the `ind` rule,

$$\frac{(\forall x : s)\ P(x)}{\bigwedge_i P(x/a_i) \wedge \bigwedge_j (\forall \overline{x} : \overline{s})\ P(x/x_{i_1}) \wedge \ldots \wedge P(x/x_{i_k}) \Rightarrow P(x/f_j(x_1, \ldots, x_{n_j}))}$$

Justification of the `ind` Rule

Why is `ind` a sound inference rule? First consider:

Lemma: Given a ground confluent, sort-decreasing, and terminating equational theory (Σ, E) with subsignature of constructors Ω, and given any Σ-equation $(\forall X)\, t = t'$, we have

$$\mathcal{T}_{\Sigma/E} \models (\forall X)\, t = t' \quad \Leftrightarrow \quad (\forall \theta : X \longrightarrow T_\Omega)\, \mathcal{T}_{\Sigma/E} \models t\theta = t'\theta.$$

Proof: Notice that for any $a : X \longrightarrow T_{\Sigma/E}$ if $x \in X_s$ we can always find $t \in a(x)$ with $t \in T_{\Sigma,s}$. Therefore we can always find a substitution $\theta : X \longrightarrow T_\Sigma$ such that for any $s \in S$ and any $x \in X_s$ we have $a(x) = [\theta(x)]$. Note also that, by the Freeness Corollary, for any $t \in T_{\Sigma(X)}$ we have $t a_{\mathcal{T}_{\Sigma/E}} = [t\theta]$. Conversely, any substitution $\theta : X \longrightarrow T_\Sigma$ determines an assignment $a : X \longrightarrow T_{\Sigma/E}$, namely, with $a(x) = [\theta(x)]$.

Justification of the `ind` Rule (II)

Therefore, we have the equivalence

$$\mathcal{T}_{\Sigma/E} \models (\forall X)\, t = t' \quad \Leftrightarrow \quad (\forall \theta : X \longrightarrow T_\Sigma)\, \mathcal{T}_{\Sigma/E} \models (\forall \emptyset)\, \overline{\theta} t = \overline{\theta} t'.$$

which proves the direction $(\Rightarrow)$ in the lemma. To prove the $(\Leftarrow)$ direction, just observe that, by the Ω being constructors, we have $C_{\Sigma/E} \subseteq T_\Omega$. Therefore, we can always choose for each assignment a a corresponding substitution θ such that for each variable x the term $\theta(x)$ is in canonical form; therefore, $\theta : X \longrightarrow T_\Omega$. q.e.d

Notice that the above Lemma easily generalizes to the modulo A case, that is, to theories $(\Sigma, E \cup A)$ with E ground confluent, sort-decreasing, and terminating modulo A and Σ preregular modulo A. Our justification of the `ind` rule in what follows works just the same for the modulo A case.

Justification of the `ind` Rule (III)

Notice that the argument of the above lemma does not depend on our formula being actually an equation: it would similarly apply to conditional equations, and even to general, universally-quantified first-order formulas.

Therefore, we have reduced the problem of proving an inductive property, $(\forall x : s)\ P(x)$, to that of proving that for all $t \in T_{\Omega,s}$ the instantiated property $P(x/t)$ holds.

Here is where structural induction steps in as a method, namely, by analyzing more closely what it means to prove something for all $t \in T_{\Omega,s}$.

Justification of the `ind` Rule (IV)

By the very inductive definition of $T_{\Omega,s}$, a term t is in $T_{\Omega,s}$ iff either:

1. $t = a$, for $a : nil \longrightarrow s'$ a constant in Ω with $s' \leq s$; or

2. there is a constructor $f : s_1 \ldots s_n \longrightarrow s'$ in Ω with $s' \leq s$ (where the sorts $s_{i_1}, \ldots, s_{i_k}$ are those among the $s_1 \ldots s_n$ such that $s_{i_j} \leq s$, $1 \leq j \leq k$) and terms $t_i \in T_{\Omega,s_i}$, $1 \leq i \leq n$ such that $t = f(t_1, \ldots, t_n)$

Therefore, given a property $P(x)$ with x a variable of sort s, if we prove that:

Justification of the `ind` Rule (V)

1. for each constant a as above, $P(x/a)$ holds; and

2. for each f as above, assuming that $P(x/t_{i_1}), \ldots, P(x/t_{i_k})$, holds then we prove that $P(x/f(t_1, \ldots, t_n))$ holds,

   (where $t_{i_j} \in T_{\Omega, s_{i_j}}$, and s_{i_j}, $1 \leq j \leq k$ are those sorts among the $s_1 \ldots s_n$ for the arguments of f such that $s_{i_j} \leq s$, $1 \leq j \leq k$)

then we have proved that $P(x/t)$ holds for all $t \in T_{\Omega,s}$ (indeed, 1–2 above amount to a proof by induction on the depth of $t \in T_{\Omega,s}$) which, by our previous lemma, shows,

$$\mathcal{T}_{\Sigma/E} \models (\forall x : s)\ P(x),$$

and therefore justifies `ind` as a sound inference rule.

Need to Check Sufficient Completeness

All this is fine, but there is a pending issue. How do we know that the declared subsignature of constructors is correct? We need to check that it is sufficiently complete, for example using the SCC tool.

Also, as discussed in the justification of the `split` rule, the user may have overlooked giving enough equations for the defined functions, and then it becomes impossible to simplify every ground term to a constructor term.

Finally, a quick glance at our proof of the lemma involved in justifying the soundness of the `ind` rule shows that the reduction of proving $P(x)$ to proving $P(x/t)$ for each $t \in T_{\Omega,s}$ works just the same under weaker assumptions than (Σ, E) ground confluent, sort-decreasing, and terminating with subsignature of constructors Ω.

Exercises

Ex.13.1 Generalize the proof justifying the soundness of the `split` rule to a considerably weaker condition than protecting. Specifically, show that the `split` rule is sound to prove inductive properties about a module $\texttt{fmod}(\Sigma, E)\texttt{endfm}$ including `BOOL` as a submodule if and only if the homomorphism

$$eval^{E'}_{\mathcal{T}_{\Sigma/E}|_{\texttt{BOOL}}} : \mathcal{T}_{\texttt{BOOL}} \longrightarrow \mathcal{T}_{\Sigma/E}|_{\texttt{BOOL}}$$

is surjective. Explain why, since `true` and `false` are the only constructors of sort `Bool`, the above surjectivity property, essential to be sure that an application of the `split` rule is correct, can be automatically checked using the Maude SCC tool under quite general assumptions on (Σ, E).

Exercises (II)

Ex.13.2 Generalize the proof justifying the soundness of the `ind` rule to a considerably weaker condition. Specifically, show that the `ind` rule is sound to prove inductive properties about a module `fmod`(Σ, E)`endfm` if and only if the structural induction scheme uses a subsignature Ω on the same sorts S such that the unique homomorphism

$$eval_{\mathcal{T}_{\Sigma/E}|_\Omega} : \mathcal{T}_\Omega \longrightarrow \mathcal{T}_{\Sigma,E}|_\Omega$$

is surjective.