Program Verification: Lecture 20

José Meseguer

Computer Science Department
University of Illinois at Urbana-Champaign

LTL Verification of Declarative Concurrent Programs

Proving that a Maude system module satisfies a property φ means proving that the corresponding initial model does:

$$\mathcal{T}_{\mathcal{R}} \models \varphi.$$

We have seen how to do this for invariants. But properties that talk about infinite behavior (e.g., fairness) require a richer logic, such as Linear Temporal Logic (LTL). Because in LTL we have a "next" operator $\bigcirc$ which talks not just about what is reachable in general, but what is reachable in one step, we will need a tighter notion of model than $\mathcal{T}_{\mathcal{R}}$. This is provided by the Kripke structure $\mathcal{K}(\mathcal{R}, k)_{\Pi}$ associated to the rewrite theory $\mathcal{R}$ with state predicates Π. So our satisfaction problem will be recast as:

$$\mathcal{K}(\mathcal{R}, k)_{\Pi}, [t] \models \varphi.$$

The Syntax of $LTL(AP)$

Given a set AP of *atomic propositions*, we define the formulae of the propositional linear temporal logic $LTL(AP)$ inductively as follows:

- **True**: $\top \in LTL(AP)$.
- **Atomic propositions**: If $p \in AP$, then $p \in LTL(AP)$.
- **Next operator**: If $\varphi \in LTL(AP)$, then $\bigcirc\varphi \in LTL(AP)$.
- **Until operator**: If $\varphi, \psi \in LTL(AP)$, then $\varphi \,\mathcal{U}\, \psi \in LTL(AP)$.
- **Boolean connectives**: If $\varphi, \psi \in LTL(AP)$, then the formulae $\neg\varphi$, and $\varphi \vee \psi$ are in $LTL(AP)$.

The Syntax of $LTL(AP)$ (II)

Other LTL connectives can be defined in terms of the above minimal set of connectives as follows:

- Other Boolean connectives:
  - **False**: $\bot = \neg\top$
  - **Conjunction**: $\varphi \wedge \psi = \neg((\neg\varphi) \vee (\neg\psi))$
  - **Implication**: $\varphi \rightarrow \psi = (\neg\varphi) \vee \psi$.

- Other temporal operators:
  - **Eventually**: $\Diamond\varphi = \top\,\mathcal{U}\,\varphi$
  - **Henceforth**: $\Box\varphi = \neg\Diamond\neg\varphi$
  - **Release**: $\varphi\,\mathcal{R}\,\psi = \neg((\neg\varphi)\,\mathcal{U}\,(\neg\psi))$
  - **Unless**: $\varphi\,\mathcal{W}\,\psi = (\varphi\,\mathcal{U}\,\psi) \vee (\Box\varphi)$
  - **Leads-to**: $\varphi \rightsquigarrow \psi = \Box(\varphi \rightarrow (\Diamond\psi))$
  - **Strong implication**: $\varphi \Rightarrow \psi = \Box(\varphi \rightarrow \psi)$
  - **Strong equivalence**: $\varphi \Leftrightarrow \psi = \Box(\varphi \leftrightarrow \psi)$.

Kripke Structures

Kripke structures are the natural models for propositional temporal logic. Essentially, a Kripke structure is a (total) unlabeled transition system to which we have added a collection of unary state predicates on its set of states.

A binary relation $R \subseteq A \times A$ on a set A is called total iff for each $a \in A$ there is at least one $a' \in A$ such that $(a, a') \in R$. If R is not total, it can be made total by defining $R^\bullet = R \cup \{(a, a) \in A^2 \mid \nexists a' \in A\ (a, a') \in R\}$.

Kripke Structures (II)

A Kripke structure is a triple $\mathcal{A} = (A, \rightarrow_{\mathcal{A}}, L)$ such that A is a set, called the set of states, $\rightarrow_{\mathcal{A}}$ is a total binary relation on A, called the transition relation, and $L : A \longrightarrow \mathcal{P}(AP)$ is a function, called the labeling function, associating to each state $a \in A$ the set $L(a)$ of those atomic propositions in AP that hold in the state a.

How can we associate a Kripke structure to a rewrite theory $\mathcal{R} = (\Sigma, E, \phi, R)$? We just need to make explicit two things: (1) the intended *kind* k of states in the signature Σ; and (2) the relevant *state predicates*, that is, the relevant set AP of atomic propositions. Having fixed k, our associated Kripke structure has as set of states those of kind k in the initial model, that is, $T_{\Sigma/E}$.

Kripke Structures (III)

The corresponding transition relation will the totalization $(\rightarrow^1_{\mathcal{R}})^\bullet$ of the one-step rewrite relation $\rightarrow^1_{\mathcal{R}}$ on $T_{\Sigma/E,k}$, where, by definition, $[t] \rightarrow^1_{\mathcal{R}} [t']$ iff there are terms $u \in [t]$ and $u' \in [t']$ and a proof $\mathcal{R} \vdash' u \rightarrow^1_{\mathcal{R}} u'$ (see Lecture 17).

If $\mathcal{R}$ satisfies the usual executability requirements we have an isomorphism $\mathcal{T}_{\mathcal{R}} \cong \mathcal{C}_{\mathcal{R}}$ and our desired Kripke structure has a much more intiuitive equivalent representation: its set of states is the set of canonical terms $C_{\Sigma/E,k}$, and its transition relation is the totalization $(\rightarrow^1_{\mathcal{C}_{\mathcal{R}}})^\bullet$ of the one-step transition relation $\rightarrow^1_{\mathcal{C}_{\mathcal{R}}}$ defined in Lecture 17.

We will explain later in this lecture how the remaining part of the Kripke structure, namely the labeling function specifying the state predicates, can also be defined.

The Semantics of $LTL(AP)$

The semantics of the temporal logic LTL is defined by means of a satisfaction relation

$$\mathcal{A}, a \models \varphi$$

between a Kripke structure $\mathcal{A}$ having AP as its atomic propositions, a state $a \in A$, and an LTL formula $\varphi \in LTL(AP)$. Specifically, $\mathcal{A}, a \models \varphi$ holds iff for each path $\pi \in Path(\mathcal{A})_a$ the path satisfaction relation

$$\mathcal{A}, \pi \models \varphi$$

holds, where we define the set $Path(\mathcal{A})$ of computation paths as the set of functions of the form $\pi : \mathbb{N} \longrightarrow A$ such that for each $n \in \mathbb{N}$, we have $\pi(n) \rightarrow_{\mathcal{A}} \pi(n+1)$ and define $Path(\mathcal{A})_a = \{\pi \in Path(\mathcal{A}) \mid \pi(0) = a\}$.

The Semantics of $LTL(AP)$ (II)

In turn we define the path satisfaction relation $\mathcal{A}, \pi \models \varphi$ in terms of the trace satisfaction relation $\tau \models \varphi$, where a trace τ is a function $\tau \in [\mathbb{N} \to \mathcal{P}(AP)]$, that is, τ is a sequence: $\tau(0)$, $\tau(1)$, $\tau(2)$, $\ldots$ $\tau(n)$, $\ldots$, with each $\tau(i) \subseteq AP$ a subset of atomic propositions. We then define $\mathcal{A}, \pi \models \varphi$ in terms of trace satisfaction by the equivalence:

$$\mathcal{A}, \pi \models \varphi \;\Leftrightarrow\; \pi; L_{\mathcal{A}} \models \varphi.$$

That is, we associate to the path π in $\mathcal{A}$ the trace $\pi; L_{\mathcal{A}}$, and then check whether φ is satisfied for that trace. Note the interesting fact that, thanks to the last equivalence, the Kripke structure $\mathcal{A}$ has dissappeared from the picture! That is, satisfaction is now defined exclusively in terms of traces in the function set $[\mathbb{N} \to \mathcal{P}(AP)]$ with no reference to $\mathcal{A}$.

The Semantics of $LTL(AP)$ (III)

Finally, we inductively define the trace satisfaction relation for any trace $\tau \in [\mathbb{N} \to \mathcal{P}(AP)]$ as follows:

- We always have $\tau \models_{LTL} \top$.
- For $p \in AP$,

$$\tau \models_{LTL} p \quad \Leftrightarrow \quad p \in \tau(0).$$

- For $\bigcirc\varphi \in LTL(AP)$,

$$\tau \models_{LTL} \bigcirc\varphi \quad \Leftrightarrow \quad s;\tau \models_{LTL} \varphi,$$

  where $s : \mathbb{N} \longrightarrow \mathbb{N}$ is the successor function.

- For $\varphi \,\mathcal{U}\, \psi \in LTL(AP)$,

$$\tau \models_{LTL} \varphi \,\mathcal{U}\, \psi \quad \Leftrightarrow$$

$$(\exists n \in \mathbb{N})\ ((s^n; \tau \models_{LTL} \psi) \wedge ((\forall m \in \mathbb{N})\ m < n \ \Rightarrow s^m; \tau \models_{LTL} \varphi)).$$

- For $\neg\varphi \in LTL(AP)$,

$$\tau \models_{LTL} \neg\varphi \quad \Leftrightarrow \quad \tau \not\models_{LTL} \varphi.$$

- For $\varphi \vee \psi \in LTL(AP)$,

$$\tau \models_{LTL} \varphi \vee \psi \quad \Leftrightarrow$$

$$\tau \models_{LTL} \varphi \quad \text{or} \quad \tau \models_{LTL} \psi.$$

The LTL Module

The LTL syntax, in a typewriter approximation of the mathematical syntax, is supported in Maude by the following LTL functional module (in the file `model-checker.maude`).

```
mod LTL is
  protecting BOOL .
  sort Formula .

  *** primitive LTL operators
  ops True False : -> Formula [ctor format (g o)] .
  op ~_ : Formula -> Formula [ctor prec 53 format (r o d)] .
  op _/\_ : Formula Formula -> Formula [comm ctor gather (E e)
                                        prec 55 format (d r o d)] .
  op _\/_ : Formula Formula -> Formula [comm ctor gather (E e)
                                        prec 59 format (d r o d)] .
```

```
op O_ : Formula -> Formula [ctor prec 53 format (r o d)] .
op _U_ : Formula Formula -> Formula [ctor prec 63 format (d r o d)] .
op _R_ : Formula Formula -> Formula [ctor prec 63 format (d r o d)] .

*** defined LTL operators
op _->_ : Formula Formula -> Formula [gather (e E) prec 65
                                                  format (d r o d)] .
op _<->_ : Formula Formula -> Formula [prec 65 format (d r o d)] .
op <>_ : Formula -> Formula [prec 53 format (r o d)] .
op []_ : Formula -> Formula [prec 53 format (r d o d)] .
op _W_ : Formula Formula -> Formula [prec 63 format (d r o d)] .
op _|->_ : Formula Formula -> Formula [prec 63 format (d r o d)] .
                                                    *** leads-to
op _=>_ : Formula Formula -> Formula [gather (e E) prec 65
                                                 format (d r o d)] .
op _<=>_ : Formula Formula -> Formula [prec 65 format (d r o d)] .

vars f g : Formula .

eq f -> g = ~ f \/ g .
eq f <-> g = (f -> g) /\ (g -> f) .
```

```
  eq <> f = True U f .
  eq [] f = False R f .
  eq f W g = (f U g) \/ [] f .
  eq f |-> g = [](f -> (<> g)) .
  eq f => g = [] (f -> g) .
  eq f <=> g = [] (f <-> g) .

  *** negative normal form
  eq ~ True = False .
  eq ~ False = True .
  eq ~ ~ f = f .
  eq ~ (f \/ g) = ~ f /\ ~ g .
  eq ~ (f /\ g) = ~ f \/ ~ g .
  eq ~ O f = O ~ f .
  eq ~(f U g) = (~ f) R (~ g) .
  eq ~(f R g) = (~ f) U (~ g) .
endfm
```

The LTL Module (II)

Note that, for the moment, no set AP of atomic propositions has been specified in the `LTL` module. We will explain in what follows how such atomic propositions are defined for a given system module `M`, and how they are added to the LTL module as a subsort `Prop` of `Formula`.

Note that the nonconstructor connectives have been defined in terms of more basic constructor connectives in the first set of equations. But since there are good reasons to put an LTL formula in negative normal form by pushing the negations next to the atomic propositions (this is specified by the second set of equations) we need to consider also the duals of the basic connectives $\top$, $\bigcirc$, $\mathcal{U}$, and $\vee$ as constructors. That is, we need to also have as constructors the dual connectives: $\bot$, $\mathcal{R}$, and $\wedge$ (note that $\bigcirc$ is self-dual).

Associating Kripke structures to Rewrite Theories

Since the models of temporal logic are Kripke structures, we need to explain how we can associate a Kripke structure to the rewrite theory specified by a Maude system module `M`.

Indeed, we associate a Kripke structure to the rewrite theory $\mathcal{R} = (\Sigma, E, \phi, R)$ specified by a Maude system module `M` by making explicit two things: (1) the intended kind k of states in the signature Σ; and (2) the relevant state predicates, that is, the relevant set AP of atomic propositions.

In general, the state predicates need not be part of the system specification and therefore they need not be specified in our system module `M`. They are typically part of the property specification.

Associating Kripke structures to Rewrite Theories (II)

This is because the state predicates need not be related to the operational semantics of `M`: they are just certain predicates about the states of the system specified by `M` that are needed to specify some properties.

Therefore, after choosing a given kind, say `[Foo]`, in `M` as our kind for states we can specify the relevant state predicates in a module `M-PREDS` which is a protecting extension of `M` according to the following general pattern:

```
mod M-PREDS is protecting M .
  including SATISFACTION .
  subsort Foo < State .
  ...
endm
```

Associating Kripke structures to Rewrite Theories (III)

Where the dots '...' indicate the part in which the syntax and semantics of the relevant state predicates is specified, as further explained in what follows. The module SATISFACTION (which is contained in the file model-checker.maude) is very simple, and has the following specification:

```
fmod SATISFACTION is
  protecting BOOL
  sorts State Prop .
  op _|=_ : State Prop -> Bool [frozen] .
endfm
```

where the sort State is unspecified. However, by importing SATISFACTION into M-PREDS and giving the subsort declaration

Associating Kripke structures to Rewrite Theories (IV)

```
subsort Foo < State .
```

all terms of sort `Foo` in `M` are also made terms of sort `State`. Note that we then have the kind identity, `[Foo]=[State]`.

The operator

```
op _|=_ : State Prop -> Bool [frozen] .
```

is crucial to define the semantics of the relevant state predicates in `M-PREDS`. Each such state predicate is declared as an operator of sort `Prop`.

In standard LTL propositional logic the set AP of atomic propositions is assumed to be a set of constants.

Associating Kripke structures to Rewrite Theories (V)

In Maude we can define parametric state predicates, that is, operators of sort `Prop` which need not be constants, but may have one or more sorts as parameter arguments. We then define the semantics of such state predicates (when the predicate holds) by appropriate equations.

We can illustrate all this by means of a simple mutual exclusion example. Suppose that our original system module `M` is the following module `MUTEX`, in which two processes, one named `a` and another named `b`, can be either waiting or in their critical section, and take turns accessing their critical section by passing each other a different token (either `$` or `*`).

Associating Kripke structures to Rewrite Theories (VI)

```
mod MUTEX is
  sorts Name Mode Proc Token Conf .
  subsorts Token Proc < Conf .
  op none : -> Conf .
  op __ : Conf Conf -> Conf [assoc comm id: none] .
  ops a b : -> Name .
  ops wait critical : -> Mode .
  op [_,_] : Name Mode -> Proc .
  ops * $ : -> Token .
  rl [a-enter] : $ [a,wait] => [a,critical] .
  rl [b-enter] : * [b,wait] => [b,critical] .
  rl [a-exit] : [a,critical] => [a,wait] * .
  rl [b-exit] : [b,critical] => [b,wait] $ .
endm
```

Associating Kripke structures to Rewrite Theories (VII)

Our obvious kind for states is the kind `[Conf]` of configurations. In order to state the desired safety and liveness properties we need state predicates telling us whether a process is waiting or is in its critical section. We can make these predicates parametric on the name of the process and define their semantics as follows:

```
mod MUTEX-PREDS is protecting MUTEX .  including SATISFACTION .
  subsort Conf < State .
  ops crit wait : Name -> Prop .
  var N : Name .
  var C : Conf .
  eq [N,critical] C |= crit(N) = true .
  eq C |= crit(N) = false [owise] .
  eq [N,wait] C |= wait(N) = true .
  eq C |= wait(N) = false [owise] .
endm
```

Associating Kripke structures to Rewrite Theories (VIII)

The above example illustrates a general method by which desired state predicates for a module `M` are defined in a protecting extension, say `M-PREDS`, of `M` which imports `SATISFACTION`.

One specifies the desired states by choosing a sort in `M` and declaring it as a subsort of `State`. One then defines the syntax of the desired state predicates as operators of sort `Prop`, and defines their semantics by means of a set of equations that specify for what states a given state predicate evaluates to `true`.

We assume that those equations, when added to those of `M`, are (ground) Church-Rosser and terminating.

Associating Kripke structures to Rewrite Theories (IX)

Since we should protect `BOOL`, it is important to make sure that satisfaction of state predicates is fully defined. This can be checked with Maude's `SCC` tool.

This means that we should give equations for when the predicates are `true` and when they are `false`. In practice, however, this often reduces to specifying when a predicate is true by means of (possibly conditional) equations of the general form,

$$t \models p(v_1, \ldots, v_n) = true \;\; \textit{if} \;\; C$$

because we can cover all the remaining cases, when it is false, with an equation

$$x : State \models p(y_1, \ldots, y_n) = false \quad [\texttt{owise}] \;\; .$$

Associating Kripke structures to Rewrite Theories (X)

In other cases, however —for example because we want to perform further reasoning using formal tools— we may fully define the true and false cases of a predicate not by using the `[owise]` attribute, but by explicit (possibly conditional) equations of the more general form,

$$t \models p(v_1, \ldots, v_n) = bexp \ \ \textit{if} \ C,$$

where $bexp$ is an arbitrary Boolean expression.

We can now associate to a system module `M` specifying a rewrite theory $\mathcal{R} = (\Sigma, E, \phi, R)$ (with a selected kind k of states and with state predicates Π defined by means of equations D in a protecting extension `M-PREDS` of `M`) a Kripke structure whose atomic predicates are specified by the set

Associating Kripke structures to Rewrite Theories (XI)

$$AP_\Pi = \{\theta(p) \mid p \in \Pi,\ \theta \text{ ground substitution}\},$$

where, by convention, we use the simplified notation $\theta(p)$ to denote the ground term $\theta(p(x_1, \ldots, x_n))$.

This defines a labeling function L_Π on the set of states $T_{\Sigma/E,k}$ assigning to each $[t] \in T_{\Sigma/E,k}$ the set of atomic propositions,

$$L_\Pi([t]) = \{\theta(p) \in AP_\Pi \mid (E \cup D) \vdash\ (\forall\, \emptyset)\ t \models \theta(p) = true\}.$$

The Kripke structure we are interested in is then

$$\mathcal{K}(\mathcal{R}, k)_\Pi = (T_{\Sigma/E,k}, (\rightarrow^1_{\mathcal{R}})^\bullet, L_\Pi)$$

Associating Kripke structures to Rewrite Theories (XII)

If $\mathcal{R}$ satisfies the usual executability requirements we have the isomorphism $\mathcal{T}_{\mathcal{R}} \cong \mathcal{C}_{\mathcal{R}}$, and $\mathcal{K}(\mathcal{R}, k)_{\Pi}$ has an isomorphic representation as the Kripke structure $(C_{\Sigma/E,k}, (\rightarrow^{1}_{\mathcal{C}_{\mathcal{R}}})^{\bullet}, L^{\mathcal{C}}_{\Pi})$, where, by definition, for each $t \in C_{\Sigma/E,k}$ we have,

$$L^{\mathcal{C}}_{\Pi}(t) = \{\theta(p) \in AP_{\Pi} \mid can_{E \cup D}(t \models \theta(p)) = true\}.$$

This is the most intituitive and computable representation for our desired Kripke structure, and indeed the one used by Maude for LTL model checking purposes. Therefore, to ensure correctness of LTL model checking in Maude it is essential to check that $\mathcal{R}$ satisfies the usual executability requirements.